

# Fantasy Sports Optimization



**First Name:** Vaidiyanathan

**Last Name:** Lalgudi Venkatesan

## Contents

<b>1</b>	<b>Problem Formulation</b>	<b>2</b>
	Parameters	2
	Variables	2
	Objective	2
<b>2</b>	<b>Team 1</b>	<b>2</b>
	Constraints	2
	Results	3
<b>3</b>	<b>Team 2</b>	<b>3</b>
	Constraints	3
	Results	4

# 1 Problem Formulation

## Parameters

$P$ : Set of Player ID (1 ... 100 as we have 100 players)

$RB_i = 1$  if Player  $i$  plays in the position Running Back or 0 otherwise;  $i \in P$

$QB_i = 1$  if Player  $i$  plays in the position Quarter Back or 0 otherwise;  $i \in P$

$WR_i = 1$  if Player  $i$  plays in the position Wide Receiver or 0 otherwise;  $i \in P$

$TE_i = 1$  if Player  $i$  plays in the position Tight End or 0 otherwise;  $i \in P$

$DST_i = 1$  if Player  $i$  plays in the position Defense or 0 otherwise;  $i \in P$

$Salary_i$  = Salary of Player  $i$ ;  $i \in P$

$Score_i$  = Score of Player  $i$ ;  $i \in P$

## Variables

$x_i = 1$  if Player  $i$  is selected to play in the team

## Objective

Maximize the Team Score: =

$$\text{Max } \sum_{i \in P} Score_i * x_i$$

# 2 Team 1

## Constraints

1. Salary Constraint: Total Salary of the selected players should be at most 50000

$$\sum_{i \in P} Salary_i * x_i \leq 50000$$

2. Individual position constraints:

There should be at least one player for each position, and the sixth player can be either RB, WR or TE.

$$\begin{aligned} \sum_{i \in P} QB_i * x_i &= 1 \\ \sum_{i \in P} RB_i * x_i &\geq 1 \\ \sum_{i \in P} WR_i * x_i &\geq 1 \\ \sum_{i \in P} TE_i * x_i &\geq 1 \\ \sum_{i \in P} DST_i * x_i &= 1 \end{aligned}$$

3. Number of Players in the team constraint:

The team should consist of exactly 6 players.

$$\sum_{i \in P} x_i = 6$$

## Results

We had to maximizing the score of the players selected, given that:

- 1) The total salary of the players should be at most 50,000.
- 2) We must have one player for each position. So, 5 players will be for each one of the positions. The 6th player could be RB, WR, or TE.
- 3) We have a total of 6 players in the team.

We got the following results using Xpress-MP:

Player ID	Player Name	Position	Salary	Player Score
14	Adam Thielen	WR	12078.7	23.95
37	Evan Engram	TE	8479.99	24.04
41	Allen Hurns	WR	7348.03	23.7
55	Ryan Griffin	QB	7094.32	24.37
71	Saints	DST	7349.52	17.61
90	Stevan Ridley	RB	5125.11	23.5

**Solution:** The maximized score that we got after solving the problem is **137.17**

### Observations:

- 1) The total salary of selected players is \$47,475, which is clearly <50,000 limit and satisfies the constraint.
- 2) We have 1 player for each position except WR for which we have 2 players. Hence, our constraint 2 has been satisfied as well.
- 3) We have a total of 6 players.

## 3 Team 2

### Constraints

1. Salary Constraint: Total Salary of the selected players should be at most 50000

$$\sum_{i \in P} \text{Salary}_i * x_i \leq 50000$$

2. Number of Players in the team constraint:

The team should consist of exactly 6 players.

$$\sum_{i \in P} x_i = 6$$

3. Team 2 should not have exactly the same players as team 1:

$$x_{14} + x_{37} + x_{41} + x_{55} + x_{71} + x_{90} \leq 5$$

4. Individual position constraints:

There should be at least one player for each position, and also have exactly 2 players in Running Back position.

$$\begin{aligned}\sum_{i \in P} QB_i * x_i &= 1 \\ \sum_{i \in P} RB_i * x_i &= 2 \\ \sum_{i \in P} WR_i * x_i &= 1 \\ \sum_{i \in P} TE_i * x_i &= 1 \\ \sum_{i \in P} DST_i * x_i &= 1\end{aligned}$$

5. Team 2 should not have the same Running back as team 1:

$$x_{90} = 0$$

6. Salary for each player should be at most 10000

$$x_i * Salary_i \leq 10000 ; \forall i \in P$$

## Results

We had to maximizing the score of the players selected, given that:

- 1) The total salary of the players should be at most 50,000. (same as that for team1).
- 2) We have a total of 6 players in the team. (same as that for team1).
- 3) **A maximum of 5 players can be same in team 2 as that in team 1.**
- 4) **We have one player for each position. So, 5 players will be for each one of the positions. We want the offense to be strong with a focus on quickly passing the football. Hence, we want that the 6<sup>th</sup> player should be a Running Back.**
- 5) **The Running back of the team 1 should not be chosen again for team 2.**
- 6) **Salary of each player should be at most 10,000 as we want to reduce the difference in salary among selected players.**

We got the following results using Xpress-MP:

Player ID	Player Name	Position	Salary	Player Score
37	Evan Engram	TE	8479.99	24.04
41	Allen Hurns	WR	7348.03	23.7
55	Ryan Griffin	QB	7094.32	24.37
71	Saints	DST	7349.52	17.61
87	Kalen Ballage	RB	7997.42	19.84
93	Rob Kelley	RB	9365.82	19.82

**Solution:** The maximized score that we got after solving the problem is **129.38**

**Observations:**

- 1) The total salary of selected players is \$47,635.1 which is <50,000 limit and satisfies the constraint 1.
- 2) We have a total of 6 players.
- 3) We have 4 common team members – players 37,41, 55,71.
- 4) We have 1 player for each position except Running Back for which we have 2 players. Hence, our constraint 4 has been satisfied as well.
- 5) Player 90, who was the Running Back for team 1 has not been chosen for team 2.
- 6) The salary for all players is under \$10,000 reducing the salary disparity among players.

## Appendix

### Code for Team 1

```
!@encoding CP1252
model ModelName
uses "mmxprs"; !gain access to the Xpress-Optimizer solver
```

```
!sample declarations section
declarations
```

```
    !Set Definitions
    p=1..100 !players
```

```
    !Decision Variables
    x: array(p) of mpvar !Should player 'i' play or not
```

```
    !Data
    RB: array(p) of integer    !Is player 'i' RB or not
    WR: array(p) of integer    !Is player 'i' WR or not
    TE: array(p) of integer    !Is player 'i' TE or not
    DST: array(p) of integer   !Is player 'i' DST or not
    QB: array(p) of integer    !Is player 'i' QB or not
    Score: array(p) of real     !Projected Score of player 'i'
    Salary: array(p) of real    !Salary of player 'i'
    Position: array(p) of string !Position of Player (Only for writing in Output)
    Name: array(p) of string    !Name of Player (Only for writing in Output)
```

```
end-declarations
```

```
initializations from 'Data.txt'
    RB
```

```

WR
QB
TE
DST
Salary
Score
Position
Name
end-initializations

forall (i in p) do
    x(i) is_binary
end-do

!Model

Obj:= sum(i in p) x(i)*Score(i)

!Salary Constraint
sum(i in p) x(i)*Salary(i) <=50000

!Total Numbers of Players in a team constraint

sum(i in p) x(i) = 6

!Individual Position Constraints
sum(i in p) x(i)*RB(i) >=1
sum(i in p) x(i)*WR(i) >=1
sum(i in p) x(i)*TE(i) >=1
sum(i in p) x(i)*DST(i)=1
sum(i in p) x(i)*QB(i)=1

maximize(Obj)

!Output
writeln("Model Solved - Optimal Objective is ",getobjval)

forall(i in p | getsol(x(i))>0) do
    writeln("x",i,"=",getsol(x(i)))
    writeln("Name","=",Name(i))
    writeln("Position","=",Position(i))
    writeln("Salary","=",Salary(i))
    writeln("Player Score","=",Score(i))
    writeln("-----")
end-do

end-model

```

## Code for Team 2

```
!@encoding CP1252
model ModelName
uses "mmxprs"; !gain access to the Xpress-Optimizer solver

!sample declarations section
declarations
    !Set Definitions
    p =1..100 !players

    !Decision Variables
    x: array(p) of mpvar !Should player 'i' play or not

    !Data
    RB: array(p) of integer    !Is player 'i' RB or not
    WR: array(p) of integer    !Is player 'i' WR or not
    TE: array(p) of integer    !Is player 'i' TE or not
    DST: array(p) of integer    !Is player 'i' DST or not
    QB: array(p) of integer    !Is player 'i' QB or not
    Score: array(p) of real      !Projected Score of player 'i'
    Salary: array(p) of real    !Salary of player 'i'
    Position: array(p) of string !Position of Player (Only for writing in Output)
    Name: array(p) of string    !Name of Player (Only for writing in Output)

end-declarations

initializations from 'Data.txt'
    RB
    WR
    QB
    TE
    DST
    Salary
    Score
    Position
    Name
end-initializations

forall (i in p) do
    x(i) is_binary
end-do

!Model
```



Obj:= sum(i in p) x(i)\*Score(i)

!Salary Constraint

sum(i in p) x(i)\*Salary(i) <=50000

!Total Numbers of Players in a team constraint

sum(i in p) x(i) = 6

!Individual Position Constraints - each position should have one player

! 2 players in running back position

sum(i in p) x(i)\*QB(i) =1

sum(i in p) x(i)\*WR(i) =1

sum(i in p) x(i)\*TE(i) =1

sum(i in p) x(i)\*DST(i)=1

!Special condition - exactly 2 RB

sum(i in p) x(i)\*RB(i)=2

!Objective not equal to first model Constraints

x(14) + x(37) + x(41) + x(55) + x(71) + x(90) <= 5

!Team 1 and 2 should have different running back

x(90) = 0

!salary for each player less than

forall ( i in p) do

    x(i)\*Salary(i) <= 10000

end-do

maximize(Obj)

!Output

writeln("Model Solved - Optimal Objective is ",getobjval)

forall(i in p | getsol(x(i))>0) do

    writeln("x",i,"=",getsol(x(i)))

    writeln("Name","=",Name(i))

    writeln("Position","=",Position(i))

    writeln("Salary","=",Salary(i))

    writeln("Player Score","=",Score(i))

    writeln("-----")

end-do

end-model